
The Reference Document (1) 1

The system is controlling cars on a bridge
between the mainland and an island

FUN-1

The number of cars on the bridge and the island
is limited

FUN-2

The bridge is one way or the other, not both at the
same time

FUN-3

The Reference Document (2) 2

The system has two traffic lights with two
colors: green and red

EQP-1

The traffic lights control the entrance to the
bridge at both ends of it

EQP-2

Cars are not supposed to pass on a red traffic
light, only on a green one

EQP-3

The Reference Document (3) 3

The system is equipped with four car sensors
each with two states: on or off

EQP-4

The sensors are used to detect the presence
of cars entering or leaving the bridge

EQP-5

Equipment 4

Bridge MainlandIsland

Our Development Approach 5

- Initial model: Limiting the number of cars (FUN 2)

- First refinement: Introducing the one way bridge (FUN 3)

- Second refinement: Introducing the traffic lights (EQP 1,2,3)

- Third refinement: Introducing the sensors (EQP 4,5)

Initial Model 6

- It is very simple

- We do not consider at all the equipment: traffic lights and sensors

- We even do not consider the bridge

- We are just interested in the pair “island-bridge”

- We are focusing on FUN-2 (limited number of cars on island-bridge)

A Situation as Seen from the Sky 7

M a i n l a n d
I s l a n d

+  b r i d g e

Two Events 8

ML_out

ML_in

8



Formalizing the State 9

constant: d

variable: n
prp0 1: d ∈ N

inv0 1: n ∈ N

inv0 2: n ≤ d

Event ML out 10

Before After

ML_out

Event ML in 11

Before

ML_in

After

Formalizing the two Events and Initialization 12

- Initially there are no cars on the Island-Bridge

- Event ML out increments the number of cars

- Event ML out decrements the number of cars

init
n := 0

ML out
n := n + 1

ML in
n := n − 1

Before-after Predicates 13

The Events

init
n := 0

ML out
n := n + 1

ML in
n := n − 1

The corresponding before-after predicates

n′ = 0 n′ = n + 1 n′ = n − 1

These representations are equivalent

Proving Invariant Preservation 14

- Given constants c with properties P (c)

- And variables v with invariants I(c, v)

- Given an event with before-after predicate v′ = E(c, v)

- We have to prove the following in order to preserve the invariant:

P (c) ∧ I(c, v) ⇒ I(c, E(c, v)) INV

Vertical Layout of the Invariant Rule 15

Properties of the constants
Invariants

⇒
Modified Invariants

P (c)
I(c, v) INV

⇒
I(c, E(c, v))

What is to be proved 16

- We have two events

ML out
n := n + 1

ML in
n := n − 1

- And two invariants

inv0 1: n ∈ N inv0 2: n ≤ d

- Thus four statements have to be proved

To be Proved for event ML out (1) 17

ML out
n := n + 1

Properties of the constants
Invariant inv0 1
Invariant inv0 2

⇒
Modified Invariant inv0 1

d ∈ N
n ∈ N
n ≤ d

⇒
n + 1 ∈ N

To be Proved for event ML out (2) 18

ML out
n := n + 1

Properties of the constants
Invariant inv0 1
Invariant inv0 2

⇒
Modified Invariant inv0 2

d ∈ N
n ∈ N
n ≤ d

⇒
n + 1 ≤ d

To be Proved for event ML in (1) 19

ML in
n := n − 1

Properties of the constants
Invariant inv0 1
Invariant inv0 2

⇒
Modified Invariant inv0 1

d ∈ N
n ∈ N
n ≤ d

⇒
n − 1 ∈ N

To be Proved for event ML in (2) 20

ML in
n := n − 1

Properties of the constants
Invariant inv0 1
Invariant inv0 2

⇒
Modified Invariant inv0 2

d ∈ N
n ∈ N
n ≤ d

⇒
n − 1 ≤ d

Summary of What is to be Proved 21

d ∈ N
n ∈ N
n ≤ d

⇒
n + 1 ∈ N

d ∈ N
n ∈ N
n ≤ d

⇒
n + 1 ≤ d

d ∈ N
n ∈ N
n ≤ d

⇒
n − 1 ∈ N

d ∈ N
n ∈ N
n ≤ d

⇒
n − 1 ≤ d

Two Proofs Fail (1) 22

d ∈ N
n ∈ N
n ≤ d

⇒
n + 1 ≤ d

- This proof fails when n = d

- We must have the additional assumption: n < d

Two Proofs Fail (2) 23

d ∈ N
n ∈ N
n ≤ d

⇒
n − 1 ∈ N

- This proof fails when n = 0

- We must have the additional assumption: 0 < n

Improving the Events 24

ML out
when

n < d
then

n := n + 1
end

ML in
when

0 < n
then

n := n − 1
end

We are adding guards to the events

The guard is the necessary condition for an event to “occur”

Adapting the Invariant Preservation Rule 25

- Given c with properties P (c) and v with invariants I(c, v)

- Given an event with guard G(c, v) and b-a predicate v′ = E(c, v)

- We modify the Invariant Preservation Rule as follows:

P (c)
I(c, v) INV

⇒
I(c, E(c, v))

;

P (c)
I(c, v)
G(c, v) INV

⇒
I(c, E(c, v))

The Modified Rule 26

Properties of the constants
Invariants
Guard of the event

⇒
Modified Invariant

P (c)
I(c, v)
G(c, v) INV

⇒
I(c, E(c, v))

Re-proving the Events: No Proofs Fail 27

d ∈ N
n ∈ N
n ≤ d
n < d

⇒
n + 1 ∈ N

d ∈ N
n ∈ N
n ≤ d
n < d

⇒
n + 1 ≤ d

d ∈ N
n ∈ N
n ≤ d
0 < n

⇒
n − 1 ∈ N

d ∈ N
n ∈ N
n ≤ d
0 < n

⇒
n − 1 ≤ d

Initialization 28

- Our system must be initialized (with no car in the island-bridge)

- The initialization event is never guarded

- It does not mention any variable in the right hand side of :=

init
n := 0

After predicate n′ = 0

Invariant Establishment Rule 29

- Given c with properties P (c) and v with invariants I(c, v)

- Given an init event with after predicate v′ = K(c)

- The Invariant Establishment Rule is the following:

Properties of the constants
⇒

Modified Invariants

P (c)
⇒ INI INV
I(c, K(c))

Applying the Invariant Establishment Rule 30

Property of the constant d
⇒

Modified Invariant

d ∈ N
⇒
0 ∈ N

d ∈ N
⇒
0 ≤ d

- Both statements hold trivially

A Missing Requirement 31

- It is possible for the system to be blocked if both guards are false

- We do not want this to happen

- We figure out that one important requirement is missing

Once started, the system should work for ever FUN-4

Proving Deadlock Freeness Rule 32

- Given c with properties P (c) and v with invariants I(c, v)

- Given the guards G1(c, v), . . . , Gm(c, v) of the events

- We have to prove the following:

P (c)
I(c, v)

⇒ DLF
G1(c, v) ∨ . . . ∨ Gm(c, v)

To be Proved 33

Properties of constants
Invariants

⇒
Disjunction of guards

d ∈ N
n ∈ N
n ≤ d

⇒
n < d ∨ 0 < n

This cannot be proved when d = 0, because then n = 0

since n ∈ N and n ≤ d. We have to add the following property:

prp0 2: 0 < d

Initial Model: Conclusion 34

- Thanks to the proofs, we discovered 3 errors

- They were corrected by:

- adding guards to both events

- adding an additional property

Proof Rules 35

- We have seen three Proof Rules:

- The Invariant Establishment Rule: INI INV

- The Invariant Preservation Rule: INV

- The Deadlock Freeness Rule (not mandatory): DLF

Proof Rules (cont’d) 36

Properties of the constants
⇒ INI INV

Modified Invariants

Properties of the constants
Invariants
Guard of the event INV

⇒
Modified Invariants

Proof Rules (cont’d) 37

Property of the constant
Invariants DLF

⇒
Disjunction of the guards

Summary of Initial Model 38

constant: d

variable: n

prp0 1: d ∈ N

prp0 2: d > 0

inv0 1: n ∈ N

inv0 2: n ≤ d

init
n := 0

ML out
when

n < d
then

n := n + 1
end

ML in
when

0 < n
then

n := n − 1
end

Our Development Approach 39

- Initial model: Limiting the number of cars (FUN 2)

- First refinement: Introducing the one way bridge (FUN 3)

- Second refinement: Introducing the traffic lights (EQP 1,2,3)

- Third refinement: Introducing the sensors (EQP 4,5)

First Refinement: Introducing a One-Way Bridge 40

- Our view of the system gets more accurate

- We introduce the bridge

- We shall refine the state

- And also add two new events

- We are focusing on FUN-3 (one way bridge)

First Refinement: Introducing a one Way Bridge 41

IL_in

I s l a n d

I s l a n d

One Way
Bridge

ML_out

IL_out ML_in

41



Introducing Three New Variables: a, b, and c 42

b

a

c

- a denotes the number of cars on bridge going to island

- b denotes the number of cars on island

- c denotes the number of cars on bridge going to mainland

Refining the State: Formalizing Variables a, b,and c 43

- Variables a, b, and c denote natural numbers

a ∈ N

b ∈ N

c ∈ N

Refining the State: Introducing New Invariants 44

- Relating the concrete state (a, b, c) to the abstract state (n)

a + b + c = n

- Formalizing the new invariant: one way bridge

a = 0 ∨ c = 0

Refining the State: Summary 45

constants: d

variables: a, b, c

inv1 1: a ∈ N

inv1 2: b ∈ N

inv1 3: c ∈ N

inv1 4: a + b + c = n

inv1 5: a = 0 ∨ c = 0

- Invariants inv1 1 to inv1 5 are called the concrete invariants

- inv1 4 glues the abstract state, n, to the concrete state, a, b, c

45



Refining Event ML out 46

ML_out

a

abstract ML out
when

n < d
then

n := n + 1
end

concrete ML out
when

a + b < d ∧ c = 0
then

a := a + 1
end

Refining Event ML in 47

ML_in

c

abstract ML in
when

0 < n
then

n := n − 1
end

concrete ML in
when

0 < c
then

c := c − 1
end

Missing Variables are not modified 48

concrete ML out
when

a + b < d
c = 0

then
a := a + 1

end

concrete ML in
when

0 < c
then

c := c − 1
end

Before-after predicates:

a′ = a + 1 ∧ b′ = b ∧ c′ = c

a′ = a ∧ b′ = b ∧ c′ = c − 1

Proving Correct Refinement: the Situation 49

Constants c with properties P (c)

Abstract variables v with abstract invariant I(c, v)

Concrete variables w with concrete invariant J(c, v, w)

Abstract event with guard G(c, v) and b-a predicate v′ = E(c, v)

Concrete event with guard H(c, w) and b-a predicate w′ = F (c, w)

Proving Correct Refinement: Invariant Refinement 50

v

w

Abstract Event

Concrete Event

J(c,v,w)

I(v) I(v’)

J(c,v’,w’)

v’=E(c,v)

w’=F(c,w)H(c,w)

G(c,v)

50



Proving Correct Refinement: Guard Strengthening 51

Properties of Constants
Abstract Invariant
Concrete Invariant
Concrete Guard

⇒
Abstract Guard

P (c)
I(c, v)
J(c, v, w) GRD REF
H(c, w)

⇒
G(c, v)

Applying Guard Strengthening to Event ML out 52

Properties of Constants

Abstract Invariant

Concrete Invariant

Concrete Guard

⇒
Abstract Guard

d ∈ N
0 < d
n ∈ N
n ≤ d
a ∈ N
b ∈ N
c ∈ N
a + b + c = n
a = 0 ∨ c = 0
a + b < d
c = 0

⇒
n < d

Applying Guard Strengthening to Event ML in 53

Properties of Constants

Abstract Invariant

Concrete Invariant

Concrete Guard
⇒

Abstract Guard

d ∈ N
0 < d
n ∈ N
n ≤ d
a ∈ N
b ∈ N
c ∈ N
a + b + c = n
a = 0 ∨ c = 0
0 < c

⇒
0 < n

Proving Correct Refinement: Invariant Refinement 54

v

w

Abstract Event

Concrete Event

J(c,v,w)

I(v) I(v’)

J(c,v’,w’)

v’=E(c,v)

w’=F(c,w)H(c,w)

G(c,v)

Invariant Refinement Rule (Special Case) 55

Properties of Constants
Abstract Invariant
Concrete Invariant
Concrete Guard
⇒
New Concrete Invariant

P (c)
I(c, v)
J(c, v, w)
H(c, w) INV REF
⇒
J(c, E(c, v), F (c, w))

Applying Invariant Refinement Rule to Event ML out
56

Properties of Constants

Abstract Invariants

Concrete Invariants

Concrete Guard

⇒
New Concrete Invariants

d ∈ N
0 < d
n ∈ N
n ≤ d
a ∈ N
b ∈ N
c ∈ N
a + b + c = n
a = 0 ∨ c = 0
a + b < d
c = 0

⇒
a + 1 ∈ N
a + 1 + b + c = n + 1
a + 1 = 0 ∨ c = 0

Applying Invariant Refinement to Event ML in 57

Properties of Constants

Abstract Invariants

Concrete Invariants

Concrete Guard
⇒
New Concrete Invariants

d ∈ N
0 < d
n ∈ N
n ≤ d
a ∈ N
b ∈ N
c ∈ N
a + b + c = n
a = 0 ∨ c = 0
0 < c

⇒
c − 1 ∈ N
a + b + c − 1 = n − 1
a = 0 ∨ c − 1 = 0

57



Refining the Initialization Event init 58

- Concrete initialization

init
a, b, c := 0, 0, 0

- Corresponding after predicate

a′ = 0 ∧ b′ = 0 ∧ c′ = 0

Refinement Rule for Initialization 59

Constants c with properties P (c)

Concrete invariant J(c, v, w)

Abstract initialization with after predicate v′ = K(c)

Concrete initialization with after predicate w′ = L(c)

Properties of the constants
⇒
Modified concrete invariants

P (c)
⇒ INI INV REF
J(c, K(c), L(c))

Applying the Initialization Refinement Rule 60

Properties of the constants

⇒
Modified concrete invariant inv1 4

(a + b + c = n)

d ∈ N
d > 0

⇒
0 + 0 + 0 = 0

Properties of the constants

⇒
Modified concrete invariant inv1 5

(a = 0 ∨ c = 0)

d ∈ N
d > 0

⇒
0 = 0 ∨ 0 = 0

New Event IL in 61

IL_in

ab

IL in
when

0 < a
then

a, b := a − 1, b + 1
end

New Event IL out 62

IL_out

b c

IL out
when

0 < b ∧ a = 0
then

b, c := b − 1, c + 1
end

Several Actions Done Together 63

IL in
when

0 < a
then

a := a − 1
b := b + 1

end

IL out
when

0 < b
a = 0

then
b := b − 1
c := c + 1

end

Before-after predicates

a′ = a + 1 ∧ b′ = b + 1 ∧ c′ = c

a′ = a ∧ b′ = b − 1 ∧ c′ = c + 1

The empty assignment skip 64

The before-after predicate of skip in the initial model

n′ = n

The before-after predicate of skip in the first refinement

a′ = a ∧ b′ = b ∧ c′ = c

Proving Correctness of New Events 65

(1) A new event must refine an implicit event, made of a skip action

(2) The new events must not diverge

- For this one has to exhibit a variant

- The variant is a natural number (could be more complicated)

- Each new event must decrease this variant

65



Event IL in Refines skip (1) 66

Properties of the constants

Abstract invariants

Concrete invariants

Concrete guards
⇒

Modified invariant inv1 4

d ∈ N
0 < d
n ∈ N
n ≤ d
a ∈ N
b ∈ N
c ∈ N
a + b + c = n
a = 0 ∨ c = 0
0 < a

⇒
a − 1 + b + 1 + c = n

Event IL in Refines skip (2) 67

Properties of the constants

Abstract invariants

Concrete invariants

Concrete guards
⇒

Modified invariant inv1 5

d ∈ N
0 < d
n ∈ N
n ≤ d
a ∈ N
b ∈ N
c ∈ N
a + b + c = n
a = 0 ∨ c = 0
0 < a

⇒
a − 1 = 0 ∨ c = 0

Non-divergence of New Events (1) 68

Properties P (c), invariants I(c, v), concrete invariant J(c, v, w)

New event with guard H(c, w)

Variant V (c, w)

Properties of the constants
Abstract invariants
Concrete invariants
Concrete guard of a new event

⇒
Variant ∈ N

P (c)
I(c, v)
J(c, v, w)
H(c, w) WFD REF1

⇒
V (c, w) ∈ N

Non-divergence of New Events (2) 69

Properties P (c), invariants I(c, v), concrete invariant J(c, v, w)

New event with guard H(c, w) and b-a predicate w′ = F (c, w)

Variant V (c, w)

Properties of cons.
Abstract invariant
Concrete invariant
Concrete guard
⇒
Modified Var. < Var.

P (c)
I(c, v)
J(c, v, w)
H(c, w) WFD REF2
⇒
V (c, F (c, w)) < V (c, w)

Proposed Variant 70

variant 1: 2 ∗ a + b

Decreasing of the Variant by Event IL in 71

IL in
when

0 < a
then

a := a − 1
b := b + 1

end

d ∈ N
0 < d
n ∈ N
n ≤ d
a ∈ N
b ∈ N
c ∈ N
a + b + c = n
a = 0 ∨ c = 0
0 < a

⇒
2 ∗ (a − 1) + b + 1 < 2 ∗ a + b

Decreasing of the Variant by Event IL out 72

IL out
when

0 < b
a = 0

then
b := b − 1
c := c + 1

end

d ∈ N
0 < d
n ∈ N
n ≤ d
a ∈ N
b ∈ N
c ∈ N
a + b + c = n
a = 0 ∨ c = 0
0 < b
a = 0

⇒
2 ∗ a + (b − 1) < 2 ∗ a + b

The Last Rule: No More Deadlock in Concrete Space
73

The Gi(c, v) are the abstract guards

The Hi(c, v) are the concrete guards

We have to prove the following Deadlock Freeness Rule

P (c)
I(c, v)
J(c, v, w)
G1(c, v) ∨ . . . ∨ Gm(c, v) DLF REF

⇒
H1(c, w) ∨ . . . ∨ Hn(c, w)

73



Applying Deadlock Freeness Rule 74

d ∈ N
0 < d
n ∈ N
n ≤ d
a ∈ N
b ∈ N
c ∈ N
a + b + c = n
a = 0 ∨ c = 0

⇒
(a + b < d ∧ c = 0) ∨
c > 0 ∨
a > 0 ∨
(b > 0 ∧ a = 0)

;

d ∈ N
0 < d
b ∈ N

⇒
b < d ∨ b > 0

74



What we Have Proved 75

- For old events:

- Strengthening of guards: rule GRD REF

- Concrete invariant preservation: rule INV REF

- For new events:

- Refining the implicit event with a skip action

- Absence of divergence: rules WFD REF1 and WFD REF2

- For all events:

- Relative deadlock freeness: rule DLF REF

75



Refinement Rules (1) 76

Properties of the constants
Abstract invariants
Concrete invariants GRD REF
Concrete guards

⇒
Abstract guards

Refinement Rules (2) 77

Properties of the constants
Abstract invariant
Concrete invariant
Concrete guard INV REF

⇒
Modified concrete invariant

Refinement Rules (3) 78

Properties of the constants
⇒ INI INV REF

Modified concrete invariants

Refinement Rules (4) 79

Properties of the constants
Abstract invariants
Concrete invariants
Concrete guard of a new event WFD REF1

⇒
Variant ∈ N

Refinement Rules (5) 80

Properties of the constants
Abstract invariants
Concrete invariants
Concrete guard of a new event WFD REF2

⇒
Modified variant < Variant

Refinement Rules (6) 81

Properties of the constants
Abstract invariants
Concrete invariants
Disjunction of abstract guards DLF REF

⇒
Disjunction of concrete guards

State of the First Refinement 82

constants: d

variables: a, b, c

inv1 1: a ∈ N

inv1 2: b ∈ N

inv1 3: c ∈ N

inv1 4: a + b + c = n

inv1 5: a = 0 ∨ c = 0

variant1: 2 ∗ a + b

82



Events of the First Refinement 83

init
a := 0
b := 0
c := 0

ML in
when

0 < c
then

c := c − 1
end

ML out
when

a + b < d
c = 0

then
a := a + 1

end

IL in
when

0 < a
then

a := a − 1
b := b + 1

end

IL out
when

0 < b
a = 0

then
b := b − 1
c := c + 1

end

Our Development Approach 84

- Initial model: Limiting the number of cars (FUN 2)

- First refinement: Introducing the one way bridge (FUN 3)

- Second refinement: Introducing the traffic lights (EQP 1,2,3)

- Third refinement: Introducing the sensors (EQP 4,5)

84



Second Refinement : Introducing Traffic Lights 85

M A I N L A N D

il_tl

ml_tl

I S L A N D

il tl ∈ {green, red} ml tl ∈ {green, red}

85



Extending the Invariant (1) 86

ml_tl

c

b

a

- A green mainland traffic light implies safe access to the bridge

86



Extending the Invariant (1) 87

ml_tl

c

b

a

- A green mainland traffic light implies safe access to the bridge

ml tl = green ⇒ c = 0 ∧ a + b < d

87



Refining Event ML out 88

a

ML_out

abstract ML out
when

c = 0 ∧ a + b < d
then

a := a + 1
end

88



Refining Event ML out 89

a

ML_out

abstract ML out
when

c = 0 ∧ a + b < d
then

a := a + 1
end

concrete ML out
when

ml tl = green
then

a := a + 1
end

89



Extending the Invariant (2) 90

c

b

a

il_tl

- A green island traffic light implies safe access to the bridge

Extending the Invariant (2) 91

c

b

a

il_tl

- A green island traffic light implies safe access to the bridge

il tl = green ⇒ a = 0 ∧ 0 < b

Refining Event IL out 92

c

IL_out

b

abstract IL out
when

a = 0 ∧ 0 < b
then

b, c := b − 1, c + 1
end

Refining Event IL out 93

c

IL_out

b

abstract IL out
when

a = 0 ∧ 0 < b
then

b, c := b − 1, c + 1
end

concrete IL out
when

il tl = green
then

b, c := b − 1, c + 1
end

93



New Events ML tl green and IL tl green 94

ML tl green
when

ml tl = red
c = 0 ∧ a + b < d

then
ml tl := green

end

IL tl green
when

il tl = red
a = 0 ∧ 0 < b

then
il tl := green

end

- Turning lights to green when proper conditions hold

94



Summary of State Refinement 95

constants: d

variables: a, b, c, ml tl, il tl

inv2 1: ml tl ∈ {red, green}

inv2 2: il tl ∈ {red, green}

inv2 3: ml tl = green ⇒ a + b < d ∧ c = 0

inv2 4: il tl = green ⇒ 0 < b ∧ a = 0

95



Superposition 96

variables: a, b, c, ml tl, il tl

- Variables a, b, and c were present in the previous refinement

- This is called superposition

- However the abstract and concrete spaces must be disjoint

- We have thus to adapt rule INV REF

96



Superposition: the situation 97

- Constant c and properties P (c)

- Abstract variables u and v and abstract invariants I(c, u, v)

- Concrete variables v and w and concrete invariants J(c, v, w)

- Abstract b-a predicate u′ = E(c, u, v) and v′ = M(c, u, v)

- Concrete guard H(c, v, w)

- Concrete b-a predicate v′ = N(c, v, w) and w′ = F (c, v, w)

97



Superposition: Change of Variable 98

- We make a change of variable: v ; v1

- We add a concrete invariant: v1 = v. Rule INV REF becomes:

Properties of constants
Abstract invariants
Concrete invariants

Concrete guards
⇒
Mod. conc inv. J(c, u, v1, w)

Mod. conc. inv. v1 = v

P (c)
I(c, u, v)
J(c, u, v1, w)
v1 = v
H(c, v1, w)
⇒
J(c, E(c, u, v), M(c, u, v),

F (c, v1, w))
M(c, u, v) = N(c, v1, w)

Superposition: Event Refinement 99

Abstract Event
when

G(c, u, v)
then

u := E(c, u, v)
v := M(c, u, v)

end

Abstract Event
when

H(c, v, w)
then

v := N(c, v, w)
w := F (c, v, w)

end

99



Superposition: Introducing a New Rule 100

- We apply the equality v1 = v

- The previous rule can be decomposed: INV REF and EQL REF

- Equality of actions on the common variables

Properties of cons.
Abstract invariant
Concrete invariant
Concrete guards

⇒
Same actions on
common variables

P (c)
I(c, u, v)
J(c, u, v, w)
H(c, v, w) EQL REF
⇒
M(c, u, v) = N(c, v, w)

Applying Rule EQL REF 101

- It is trivial since the actions are the same

(abstract )ML out
when

c = 0 ∧ a + b < d
then

a := a + 1
end

(concrete )ML out
when

ml tl = green
then

a := a + 1
end

- Same thing for refinement of event IL out

Proving Preservation of inv2 4 by Event ML out 102

a ∈ N
il tl = green ⇒ a = 0 ∧ 0 < b
ml tl = green

⇒
il tl = green ⇒ a + 1 = 0 ∧ 0 < b

It cannot be proved when il tl = green since it reduces to

a ∈ N
a = 0

⇒
a + 1 = 0

Proving Preservation of inv2 3 by Event IL out 103

c ∈ N
ml tl = green ⇒ c = 0 ∧ a + b < d
il tl = green

⇒
ml tl = green ⇒ c + 1 = 0 ∧ a + b − 1 < d

It cannot be proved when ml tl = green since it reduces to

c ∈ N
c = 0

⇒
c + 1 = 0

Adding a New Invariant 104

Proof failures shows that both lights cannot be green at the same time

An obvious fact that we had forgotten to state

inv2 5: ml tl = red ∨ il tl = red

Going back to the Requirements Document 105

inv2 5: ml tl = red ∨ il tl = red

This could have been deduced from these requirements

The bridge is one way or the other, not both at the
same time FUN-3

Cars are not supposed to pass on a red traffic
light, only on a green one EQP-3

105



Correcting the New Events 106

But the new invariant inv2 5 is not preserved by the new events

Unless we correct them as follows:

ML tl green
when

ml tl = red
a + b < d
c = 0

then
ml tl := green
il tl := red

end

IL tl green
when

il tl = red
0 < b
a = 0

then
il tl := green
ml tl := red

end

Proving Preservation of inv2 3 by Event ML out 107

ml tl = green ⇒ c = 0 ∧ a + b < d
ml tl = green

⇒
ml tl = green ⇒ c = 0 ∧ a + 1 + b < d

It cannot be proved when a+1+ b = d unless il tl becomes equal

to red

Solution: Splitting Event ML out 108

ML out 1
when

ml tl = green
a + b + 1 6= d

then
a := a + 1

end

ML out 2
when

ml tl = green
a + b + 1 = d

then
a := a + 1
ml tl := red

end

Proving Preservation of inv2 4 by Event IL out 109

il tl = green ⇒ a = 0 ∧ 0 < b
il tl = green

⇒
il tl = green ⇒ a = 0 ∧ 0 < b − 1

It cannot be proved when b = 1 unless il tl becomes equal to red

Solution: Splitting Event IL out 110

IL out 1
when

il tl = green
b 6= 1

then
b := b − 1
c := c + 1

end

IL out 2
when

il tl = green
b = 1

then
b := b − 1
c := c + 1
il tl := red

end

Summary of the Proof Situation 111

- Correct event refinement: OK

- Absence of divergence of new events: FAILURE

- Absence of deadlock: ?

111



Divergence of the New Events 112

ML tl green
when

ml tl = red
a + b < d
c = 0

then
ml tl := green
il tl := red

end

IL tl green
when

il tl = red
0 < b
a = 0

then
il tl := green
ml tl := red

end

When a and c are both equal to 0 and b is positive, then both events

are always alternatively enabled

The lights can change colors very rapidly

ML tl green and IL tl green can run for ever 113

IL ML

ml_tl

il_tl

a=0

c=0

ML tl green and IL tl green can run for ever 114

IL ML

ml_tl

il_tl

a=0

c=0

ML tl green and IL tl green can run for ever 115

IL ML

ml_tl

il_tl

a=0

c=0

ML tl green and IL tl green can run for ever 116

IL ML

ml_tl

il_tl

a=0

c=0

ML tl green and IL tl green can run for ever 117

IL ML

ml_tl

il_tl

a=0

c=0

ML tl green and IL tl green can run for ever 118

IL ML

ml_tl

il_tl

a=0

c=0

Solution 119

- Allowing each light to turn green only when at least one car

has passed in the other direction

- For this, we introduce two additional variables:

inv2 6: ml pass ∈ {0, 1}

inv2 7: il pass ∈ {0, 1}

Modifying Events ML out 1 and ML out 2 120

ML out 1
when

ml tl = green
a + b + 1 6= d

then
a := a + 1
ml pass := 1

end

ML out 2
when

ml tl = green
a + b + 1 = d

then
a := a + 1
ml tl := red
ml pass := 1

end

Modifying Events ML out 1 and ML out 2 121

IL out 1
when

il tl = green
b 6= 1

then
b := b − 1
c := c + 1
il pass := 1

end

IL out 2
when

il tl = green
b = 1

then
b := b − 1
c := c + 1
il tl := red
il pass := 1

end

Modifying Events ML tl green and IL tl green 122

ML tl green
when

ml tl = red
a + b < d
c = 0
il pass = 1

then
ml tl := green
il tl := red
ml pass := 0

end

IL tl green
when

il tl = red
0 < b
a = 0
ml pass = 1

then
il tl := green
ml tl := red
il pass := 0

end

Proving Absence of Divergence 123

We exhibit the following variant

variant 2: ml pass + il pass

To be Proved 124

ml tl = red
a + b < d
c = 0
il pass = 1

⇒
il pass + 0 <
ml pass + il pass

il tl = red
b > 0
a = 0
ml pass = 1

⇒
ml pass + 0 <
ml pass + il pass

This cannot be proved. This suggests the following invariants:

inv2 8: ml tl = red ⇒ ml pass = 1

inv2 9: il tl = red ⇒ il pass = 1

124



No Deadlock (1) 125

0 < d
ml tl ∈ {red, green}
il tl ∈ {red, green}
ml pass ∈ {0, 1}
il pass ∈ {0, 1}
a ∈ N
b ∈ N
c ∈ N
ml tl = red ⇒ ml pass = 1
il tl = red ⇒ il pass = 1

⇒
(ml tl = red ∧ a + b < d ∧ c = 0 ∧ il pass = 1) ∨
(il tl = red ∧ a = 0 ∧ b > 0 ∧ ml pass = 1) ∨
ml tl = green ∨ il tl = green ∨ a > 0 ∨ c > 0

No Deadlock (2) 126

The previous statement reduces to the following, which is true

0 < d
a ∈ N
b ∈ N
c ∈ N

⇒
(a + b < d ∧ c = 0) ∨
(a = 0 ∧ b > 0) ∨
a > 0 ∨
c > 0

;

0 < d
b ∈ N

⇒
b < d ∨ b > 0

Second Refinement: Conclusion 127

- Thanks to the proofs:

- We discovered 4 errors

- We introduced several additional invariants

- We corrected 4 events

- We introduced 2 more variables

Conclusion: we Introduced the Superposition Rule 128

Properties of constants
Abstract invariants
Concrete invariants EQL REF
Concrete guards

⇒
Same actions on common variables

Summary of Second Refinement: the State (1) 129

constants: d

variables: a, b, c,
ml tl, il tl,
ml pass, il pass

inv2 1: ml tl ∈ {red, green}

inv2 2: il tl ∈ {red, green}

inv2 3: ml tl = 1 ⇒ a + b < d ∧ c = 0

inv2 4: il tl = 1 ⇒ 0 < b ∧ a = 0

Summary of Second Refinement: the State (2) 130

inv2 5: ml tl = red ∨ il tl = red

inv2 6: ml pass ∈ {0, 1}

inv2 7: il pass ∈ {0, 1}

inv2 8: ml tl = red ⇒ ml pass = 1

inv2 9: il tl = red ⇒ il pass = 1

variant2: ml pass + il pass

Summary of Second Refinement: the Event (1) 131

ML out 1
when

ml tl = green
a + b + 1 6= d

then
a := a + 1
ml pass := 1

end

ML out 2
when

ml tl = green
a + b + 1 = d

then
a := a + 1
ml pass := 1
ml tl := red

end

Summary of Second Refinement: the Event (2) 132

IL out 1
when

il tl = green
b 6= 1

then
b := b − 1
c := c + 1
il pass := 1

end

IL out 2
when

il tl = green
b = 1

then
b := b − 1
c := c + 1
il pass := 1
il tl := red

end

Summary of Second Refinement: the Event (3) 133

ML tl green
when

ml tl = red
a + b < d
c = 0
il pass = 1

then
ml tl := green
il tl := red
ml pass := 0

end

IL tl green
when

il tl = red
0 < b
a = 0
ml pass = 1

then
il tl := green
ml tl := red
il pass := 0

end

Summary of Second Refinement: the Event (4) 134

- These events are identical to their abstract versions

ML in
when

0 < c
then

c := c − 1
end

IL in
when

0 < a
then

a := a − 1
b := b + 1

end

Our Development Approach 135

- Initial model: Limiting the number of cars (FUN 2)

- First refinement: Introducing the one way bridge (FUN 3)

- Second refinement: Introducing the traffic lights (EQP 1,2,3)

- Third refinement: Introducing the sensors (EQP 4,5)

Third Refinement: Adding Car Sensors 136

ML

IL

il_in_sr

ml_tl

il_tl

il_out_sr

ml_in_sr

ml_out_sr

Extending the State 137

constants: d

variables: a, b, c,
ml tl,
il tl,
ml pass,
il pass,
ml out sr,
ml in sr,
il out sr,
il in sr

inv3 1: ml out sr ∈ {0, 1}

inv3 2: ml in sr ∈ {0, 1}

inv3 3: il out sr ∈ {0, 1}

inv3 4: il in sr ∈ {0, 1}

Extending the State 138

inv3 5: il in sr = 1 ⇒ a > 0

inv3 6: il out sr = 1 ⇒ b > 0

inv3 7: ml in sr = 1 ⇒ c > 0

Refining Abstract Events (1) 139

ML out 1
when

ml tl = green
a + b + 1 6= d
ml out sr = 1

then
a := a + 1
ml pass := 1
ml out sr := 0

end

ML out 2
when

ml tl = green
a + b + 1 = d
ml out sr = 1

then
a := a + 1
ml tl := red
ml pass := 1
ml out sr := 0

end

139



Refining Abstract Events (2) 140

IL out 1
when

il tl = green
b 6= 1
il out sr = 1

then
b := b − 1
c := c + 1
il pass := 1
il out sr := 0

end

IL out 2
when

il tl = green
b = 1
il out sr = 1

then
b := b − 1
c := c + 1
il tl := red
il pass := 1
il out sr := 0

end

Refining Abstract Events (3) 141

ML in
when

ml in sr = 1
then

c := c − 1
ml in sr := 0

end

IL in
when

il in sr = 1
then

a := a − 1
b := b + 1
il in sr := 0

end

Refining Abstract Events (4) 142

ML tl green
when

ml tl = red
a + b < d
c = 0
il pass = 1
ml out sr = 1

then
ml tl := green
il tl := red
ml pass := 0

end

IL tl green
when

il tl = red
a = 0
ml pass = 1
il out sr = 1

then
il tl := green
ml tl := red
il pass := 0

end

Adding New Events (1) 143

ML out arr
when

ml out sr = 0
then

ml out sr := 1
end

ML in arr
when

ml in sr = 0
c > 0

then
ml in sr := 1

end

Adding New Events (1) 144

IL in arr
when

il in sr = 0
a > 0

then
il in sr := 1

end

IL out arr
when

il out sr = 0
b > 0

then
il out sr := 1

end

144



Final Structure of the Controller 145

il_out_sr

il_in_sr

ml_out_sr

ml_in_sr

ml_tlil_tl

Constant:  d

il_pass,
ml_pass

12 Transitions

Variables:  a, b, c,

A Short Demonstration 146

- The constant d is set to 4

- We have 5 cars named 1, 2, 3, 4, and 5

- The 12 transitions have been translated in an obvious way.

- Transitions ML out and IL out have been randomized

- This simulation worked first time

Questions on Proving 147

- What is to be systematically proved?

- Invariant preservation

- Correct refinements of transitions

- No divergence of new transitions

- No deadlock introduced in refinements

- When are these proofs done?

Questions on Proving (cont’d) 148

- Who states what is to be proved?

- An automatic tool: the Proof Obligation Generator

- Who is going to perform these proofs?

- An automatic tool: the Prover

- Sometimes helped by the Engineer (interactive proving)

About Tools 149

- Three basic tools:

- Proof Obligation Generator

- Prover

- Model translators into Hardware or Software languages

- These tools are embedded into a Development Data Base

- Such tools already exist: Atelier B and Click’n’Prove

Summary of Proofs on Example 150

- This development required 125 proofs

- Only 4 of them were interactive (easy)

- 97% proved automatically

Summary of Mathematical Notations (1) 151

P ∧ Q conjunction

P ∨ Q disjunction

P ⇒ Q implication

¬ P negation

x ∈ S set membership operator

Summary of Mathematical Notations (2) 152

N set of Natural Numbers: {0, 1, 2, 3, . . .}

Z set of Integers: {0, 1, −1, 2, −2, . . .}

{a, b, . . .} set defined in extension

a + b addition of a and b

a − b subtraction of a and b

Summary of Mathematical Notations (3) 153

a ∗ b product of a and b

a = b equality relation

a ≤ b smaller than or equal relation

a < b smaller than relation

Invariant Establishment Proof Rule 154

- For the init event in the initial model

Properties of the constants
⇒ INI INV

Modified Invariants

Invariant Preservation Proof Rule 155

- For other events in the initial model

Properties of the constants
Invariants
Guard of the event INV

⇒
Modified Invariants

Deadlock Freeness Rule 156

- This rule is not mandatory

Property of the constant
Invariants DLF

⇒
Disjunction of the guards

Refinement Rules (1): Guard Strengthening 157

- For old events only

Properties of the constants
Abstract invariants
Concrete invariants GRD REF
Concrete guards

⇒
Abstract guards

Refinement Rules (2): Invariant Establishment 158

- For init event only

Properties of the constants
⇒ INI INV REF

Modified concrete invariants

Refinement Rules (3): Invariant Preservation 159

- For all events (except init)

- New events refine an implicit non-guarded event with skip action

Properties of the constants
Abstract invariant
Concrete invariant
Concrete guard INV REF

⇒
Modified concrete invariant

Refinement Rules (4): Non-divergence of New Events
160

- For new events only

Properties of the constants
Abstract invariants
Concrete invariants
Concrete guard of a new event WFD REF1

⇒
Variant ∈ N

Refinement Rules (5): Non-divergence of New Events
161

- For new events only

Properties of the constants
Abstract invariants
Concrete invariants
Disj. of abs. guards WFD REF2

⇒
Disj. of conc. guards

Refinement Rules (6): Relative Deadlock Freeness 162

- Global proof rule

Properties of the constants
Abstract invariants
Concrete invariants
Disjunction of abstract guards DLF REF

⇒
Disjunction of concrete guards

Refinement Rules (7) 163

- For old events (in case of superposition)

Properties of constants
Abstract invariants
Concrete invariants EQL REF
Concrete guards

⇒
Same actions on common variables

